COMPUTER ANALYSIS APPLIED TO THE METHOD OF VIRTUAL WORK

A.F.S. Diwan Ph.D

Structural Engineering Departerment, Faculty of Engineering
Alexandria university, Alexandria, Egypt.

1. INTRODUCTION

The method of virtual work is one of the most able tools in the analysis of structures, determinate or indeterminate, under external action, whether loads, temperature change, yielding of supports, or influence lines. When using the method of virtual work to the analysis of indeterminate structures. it is necessary to solve a number of simultaneous equations, the coefficients of which may be called (displacement coefficients). The determination of these constants, and then the solution of these equations limited the use of the method of virtual work in the analysis of highly redundant frames. For frames n-times statically indeterminate (nxn) equations need to be formed with necessary displacement coefts. $(n^2+n)/2$, and then solved. This is why this method was limited to frames 3-times statically indeterminate or less.

So for, the methods used with the computer in the analysis of indeterminate frames are two; namely the displacement method, and the force method. This paper gives the solution using the direct method of virtual work. It will be seen that this method of analysis will be rather simpler than the two methods already used in the computer.

2. DISPLACEMENT COEFFICIENTS

The displacement coeffitionts needed for the solution in a frame n-times statically indeterminate are of the following types:

(a)
$$d_{10} = \int M_o M_1 \, dL/EI \quad (1\text{-a})$$
$$dno = \int M_o M_n \, dL/EI \quad (1\text{-n})$$

and:

(b)
$$d_{11} = \int M_1^2 \, dL/E1 \quad (2\text{-a})$$
$$d_{mn} = \int M_m M_n \, dL/E1 \; (2\text{- } (n^2+n)/2)$$

Notice that:

1. *The integration covers all members*
2. M_o, M_n represent B.Ms. in the statically determinate main-system (M.S.) due to M_o, M_n diagrams.
3. The effects of the normal forces N, and shearing forces Q are not included, but they can be added.

We better replace the term dL/Ei by $d\bar{L}$; elastic length element:

For frames without arches; or with unloaded arches, the M.s. may be obtained by introducing cut sections, internal or external hinges movable supports or free ends. For pannels with arches, the M.s may be obtained by introducing a movable support on one end and a hinge in the other side. The arches used have a second-degree equation for their axes, with constant $d\bar{L}$ for the rib. The acting loads thereon are uniform loads. The B.Ms. in these arches in the M.S for M_o and $_n$ will then be second degree parabolas.

3. ADDITIONAL MOMENT DIAGRAMS.

The B.M.D. for the various parts in the system will consist of linear diagrams to which, additional moment diagrams due to external loads thereon, (or to the presence of arched 2-nd degree parabolas) will have to be added. The average ordinate in these parabolas depend on the type of the loads therein, as given in Table (1).
Notice that the average value does not depend on the elastic length

Sign Convention

In the integrations (1), (2), positive values correspond to moments with the same sign. For vertical members we may consider moments producing tension to the left of the member as positive and those producing tension to the right as negative, except the first column to the left where moments producing tension to left (outside the forme will have to be negative. For other members, moments producing tension inside the forme will be considered positive moments while those producing tension outside the forme will then be negative.

Table 1. Average moment A (I)

Nº	Loading	A
1	 Uniform	$pL^2/12$
2		$pL^2/32$
3		$5pL^2/96$
4	 2nd par.	$pL^2/15$
5	 Sine	$2pL^2/\pi^3$
6		$pL^2/24$

L = span length p = max intensity of lood
G = ratio of dist of c.g. of M. Diag. to L = .50 for cases 1-5 & = 8/15 from left for case 6.

Steps of solution

The computer program is given in appendix (I) at the end of this paper. It gives the final B.Ms. in frames with or without parabolic arches under the given loads. The program consists of a main part followed by a subroutine necessary to solve the necessary simultaneous equations. It is written in BASIC language. When saved in the computer it may be saved as: Save "VWP. BAS", A. This allows the program to be merged with the Data-program which is saved separately with numbers between the main program and its subroutine. To enable the user to solve many example, a diagram "DATAVWP" may be saved and then called to fit in the required data for the specific problem to be solved. This program, if required may be saved as "DATAVWPn" where n refers to the example. To this last program we may add the original program by this order: MERGE "VWP". If this final program is now run, the final bending moments will be given either on the screen, or be printed.

EXAMPLES.

Example (1)

Figure (1-a) shows the given frame, which is 7- times statically indeterminate. It also shows the acting outside loads. The number of acting moments in the frame, denoted by N will be 20. Figure (1-b) shows the chosen statically determinate system (M.S), and the adopted sign for the moments, and the sections where they act.

It also shows the redundant moments Figure. (1-c) gives the determinate moments M_o, and the rest of diagrams (d) to (i) give the M_n diagrams. The moments not shown in the Figures have zero values. We may now call the (DATAVWP) diagram and fill in the necessary data to get the data diagram, which may be called "DATAVWO2" as shown. To this program we can now "Merge" the "VWP" program by giving the following order, MERGE "VWP" to get the full program which when (RUN) gives the final B. Moments. Theses moments are also shown. If required we can "Save" The (DATAVWPI) program by the ordinary method.



Figure 2.

Example (2)

This example is shown in Figure (2), from (2-a) to (2-i). The frame there has there parabolas, each being a second degree parabol with const (dL/EI) value. In this case R = 7, N = 14. From Figures (2) we can obtain the nessary data, which enables to get the data program "DATAVWP1". This program is given in the appendices. if then we (MERGE) this program with the "VWP" program we get the full program. Running this program we can get the final moments on the screen or printer as required.

Case of several loadings

Program "VWPB" is given in appendix (3). This program enables to solve these frames subjected to several cases of external loadings. B refers to these various cases of loadings and is equal to their number. Notice that the moments M and the maximum values w1 therein will not change, while the moments Mo and the average values A will correspond to the several cases of loading B. Similarly T and U at the end of the program "DATAVWPB" which is given in Appendix (4), which is applicable to all such problems.

Example (3)

This example has the same dimensions given before in example 2. The acting loads are here devided to three cases of loading, namely the horizontal load, then the vertical uniform loads on the first two parabolas, and finally both loads acting together. B will then be equal to three. This enables to give the required data which is shown in the program "DATVWPBI" given in the paper. The data for M, w_1 and L are the same as in example (2). To this program MERGE the (VWPB) to get the full program necessary to give (when Run) the final moments
program necessary to give (when Run) the final moments
for all three cases of loading



(1-d) M1 - Diag.   1-j - M7 - Diag.



(1-e) M2_Diag.   (1-f) M6_Diag.



(1-f) M3_Diag.



(1-g) M4. Diag.



h- M5 - Diag. (1-h)

EXAMPLE (2)



2-a- Given Frame & Loads



2-b- Un-Known Redundant Values



2-C. Mo - Diag.



2-d- M1 - Diag.



2-e- M2. Diag.



2-f M3 - Diag.



2-g M4 - Diag.



2-h M5 - Diag.



2-i M6 - Diag.



2-j M7. Diag.

```
10 PRINT TAB(4)," NAME OF PROGRAM  ----   VW  ---- " : PRINT
20 PRINT TAB(4)," THIS PROGRAM GIVES THE SOLUTION BY  VW" : PRINT
30 PRINT "--------------------------------------------------" : PRINT  : PRINT
40 REM  R = NOM. OF REDUNDANT VALUES
50 REM  N = NOM. OF MOMENTS AT THE ENDS OF MEMBERS
60 REM  M0 = MOMENTS IN THE M.S  DUE TO GIVEN LOADS
70 REM  M(I) = MOMENTS DUE TO X(I) = 1
90 REM  G GIVES THE POSITION OF THE C.G. OF ADDITIONAL M0 DIAG.
100 REM  A = AVERAGE MOMENT IN THE ADDITIONAL M0 DIAG.
110 REM  D(I) = DISPLACEMENT COEFS. FROM M0 & M(I) DIAGS.
120 REM  S(I,J) = .................... FROM APPROPRIATE M(I,J) DIAGS.
140 REM T(I) = ADDITIONAL VALUE TO S(I,J) DUE TO TIE IF ANY
150 REM U(I) = ................... D(I)  DUE TO SETTLEMENT IF ANY
180  READ R , N
190  PRINT " R = " , R  : PRINT " N = " , N  : PRINT
210        DIM L(N/2) , M0(N) , A(N/2) , G(N/2)
220       DIM D(R) , M(R,N) , S(R,R), C(R,R) , BM(N), Z(R,2*R)
240 PRINT  TAB(6)" 1 - ELASTIC LENGTHES L(I) "  :  PRINT
250      FOR I = 1 TO (N /2)
260 READ L(I) : PRINT TAB(4) L(I) : NEXT I : PRINT
280  PRINT TAB(6)" 2 - MOMENTS M0 IN  M.S. "  :  PRINT
290   FOR I = 1 TO N
300 READ M0(I) : PRINT TAB(4) M0(I) : NEXT  I : PRINT
320  PRINT TAB(6) " 3 - AVERAGE ADDITIONAL AREAS A(I) IN M0 DIAG." : PRINT
330  FOR I = 1 TO N / 2
340  READ A(I): PRINT A(I) :NEXT I : PRINT
360 PRINT TAB(6)" 4 - POSITION OF CENTROID  G FOR CG. OF A " : PRINT
370  FOR  I = 1 TO N / 2
380  READ G(I): PRINT  G(I) : NEXT  I : PRINT
400 PRINT  TAB(6)" 5 - MOMENTS M(I,J) DUE TO M(I) = 1 " :  PRINT
410  FOR I = 1 TO R
420   FOR J = 1 TO N
430  READ M(I,J) : PRINT M(I,J), : NEXT J
440   PRINT  : NEXT  I : PRINT
460 PRINT TAB(6) " 6 - ADDITIONAL DISP. COEF. U(I) DUE TO SETTLEMENT " : PRINT
470         FOR I = 1 TO R
480  READ U(I) : PRINT TAB(4) U(I) : NEXT I : PRINT
500 PRINT TAB(6) " 7 - ADDITIONAL DISP. COEF. T(I) DUE TO TIES      " : PRINT
510  FOR I = 1 TO R
520   READ T(I) : PRINT  T(I) : NEXT I : PRINT
550 PRINT TAB(6) " 8 - DISP. COIFS. D(I) DUE TO M0 & M1 DIAGS. " : PRINT
560  FOR I = 1 TO R
570   D(I) = 0
580   FOR K = 1 TO (N - 1)  STEP 2
590  D(I) = D(I) + L((K+1)/2) * ( M0(K)*M(I,K) /3  +  M0(K+1) * M(I,(K+1))/3 )
600 D(I)= D(I) +L((K+1)/2) *( M0(K)* M(I,(K+1))/6  + M0(K+1)* M(I,K )/6)
610 D(I)= D(I) +L((K+1)/2)* A((K+1)/2)*M(I,(K+1))* G((K+1)/2)
```

```
620 D(I) =D(I) + L((K+1)/2) * A((K+1)/2) * (1-G((K+1)/2)) * M(I,K)
640    NEXT K
650  D(I) = D(I) + U(I)
660      PRINT D(I) : NEXT I : PRINT
700 PRINT TAB(6)"9 - DISP. COEFS. S(I,J)  DUE TO M(I,J) " : PRINT
710  FOR I = 1 TO R
720    FOR J = 1  TO R
730      S(I,J) = 0
740   FOR K = 1 TO (N-1) STEP 2
750 S(I,J) = S(I,J) + L((K+1)/2)  *(M(I,K)*M(J,K)/3 + M(I,(K+1))*M(J,(K+1))/3)
760 S(I,J) = S(I,J) + L((K+1)/2)* (M(I,K)*M(J,(K+1))/6 + M(I,(K+1))* M(J,K)/6)
770     NEXT K
780      IF J<>I THEN GOTO 800
790   S(I,J) =   S(I,J) + T(I)
800    NEXT J   :  NEXT I
820   FOR I = 1 TO R
830    FOR J = 1 TO R
840  PRINT S(I,J), : NEXT  J : PRINT
850   NEXT I : PRINT  : PRINT
880 PRINT TAB(6) " 10 - FLEXIBILITY MATRIX V = INVERS OF S " :  PRINT
910       FOR I = 1 TO R
920         FOR J = 1 TO R
930           Z(I,J) = S(I,J)
940                        NEXT  J  :          NEXT I: PRINT
950     IF R = 1 THEN  V(1,1) = 1 / S(1,1)  : PRINT V(1,1) :GOTO 1000
980  GOSUB 10000
1000  PRINT TAB(6)" 11 - REDUNDANT VALUES X (R) " : PRINT
1110    FOR I = 1 TO  R
1120        X(I) = 0
1140    FOR K = 1 TO   R
1150       X(I) = X(I) -V(I,K) * D(K)
1160         NEXT K : NEXT  I
1170   FOR I = 1 TO R
1180     PRINT TAB(4) X(I) :NEXT I : PRINT
1200     PRINT TAB(6)" 11 - FINAL B.MOMENTS BM     " : PRINT
1210  FOR I = 1 TO N
1220       BM(I) = 0
1230    FOR J = 1 TO R
1240     BM(I) = BM(I) + X(J) * M(J,I)
1250         NEXT  J
1260     BM(I) = BM(I) + M0(I)
1270      NEXT  I
1272 INPUT "YOU WANT SCREEN (1) OR PRINTER (2) , OR END (3) " , SP
1274    ON SP GOTO 1280 , 1360 , 1500
1280  FOR I = 1 TO (N-1) STEP 2
1290   PRINT TAB(4) BM(I) , BM(I+1)  :  PRINT
1310     NEXT I : PRINT
1320   PRINT TAB(6) " THIS IS ALL NOW " : PRINT
```

```
1340     GOTO 1272
1360 FOR I = 1 TO (N-1)  STEP 2
1370  LPRINT TAB(4) BM(I) , BM(I+1) : LPRINT
1380   NEXT I  :  LPRINT
1400  LPRINT TAB(4) " THAT  IS  ALL  NOW " :  LPRINT
1500     END
10000    FOR I = 1 TO R
10010      FOR J = (R+1) TO  2 * R
10020       Z(I,J) = 0
10030        NEXT J
10035     Z(I , R+I) = 1
10040        NEXT I
10050    P = 0
10060       P = P + 1
10070    C(P,P) = Z(P,P)
10080      FOR    J = 1 TO (2 * R)
10090    Z(P,J) = Z(P,J) / C(P,P)
10100        NEXT J
10110    FOR I = (P + 1)  TO  R
10120       C(I,P) = Z(I,P)
10130    FOR J = P  TO 2 * R
10140     Z(I,J) = Z(I,J) - C(I,P) * Z(P,J)
10150          NEXT J
10160            NEXT I
10170         IF P = 1  THEN 10060
10180     FOR I = 1 TO  (P-1)
10190       C(I,P) = Z(I,P)
10200    FOR J = P TO (2 * R)
10210     Z(I,J) = Z(I,J) - C(I,P) * Z(P,J)
10220        NEXT  J
10230          NEXT I
10240    IF P = R THEN GOTO 10260
10250      GOTO  10060
10260  FOR I = 1 TO R
10270    FOR J = 1 TO R
10280    V(I,J) = Z(I,R+J)
10290  PRINT V(I,J), : NEXT J
10300     PRINT   : NEXT  I : PRINT
10320        RETURN
```

```
10 PRINT TAB(4)," NAME OF PROGRAM  ----   VWP ---- " : PRINT
15 PRINT TAB(4)," THIS PROGRAM GIVES THE SOLUTION BY  VW" : PRINT
20 PRINT TAB(6)" ---FOR FRAMES WITH PARABOLIC ARCHES  --- " : PRINT
30 PRINT "--------------------------------------------------" : PRINT  : PRINT
40 REM  R = NOM. OF REDUNDANT VALUES
50 REM  N = NOM. OF MOMENTS AT THE ENDS OF MEMBERS
60 REM  M0 = MOMENTS IN THE M.S  DUE TO GIVEN LOADS
70 REM  M(I) = MOMENTS DUE TO X(I) = 1
90 REM  G GIVES THE POSITION OF THE C.G. OF ADDITIONAL M0 DIAG.
100 REM  A = AVERAGE MOMENT IN THE ADDITIONAL M0 DIAG.
105 REM W1 = MAX. ORDINATE IN PARABOLA IN M1 - DIAGS. "
110 REM  D(I) = DISPLACEMENT COEFS. FROM M0 & M(I) DIAGS.
120 REM  S(I,J) = .................... FROM APPROPRIATE M(I,J) DIAGS.
140 REM T(I) = ADDITIONAL VALUE TO S(I,J) DUE TO TIE IF ANY
150 REM U(I) = .................. D(I)  DUE TO SETTLEMENT IF ANY
180   READ R , N
190  PRINT " R = " , R  : PRINT " N = " , N  : PRINT
210       DIM L(N/2) , M0(N) , A(N/2) , G(N/2)
220      DIM D(R) , M(R,N) , S(R,R), C(R,R),Z(R,2*R), V(R,R)
230  PRINT
240 PRINT  TAB(6)" 1 - ELASTIC LENGTHES L(I) "  :  PRINT
250        FOR I = 1 TO (N /2)
260 READ L(I) : PRINT TAB(4) L(I) : NEXT I : PRINT
280  PRINT TAB(6)" 2 - MOMENTS M0 IN  M.S. "  :  PRINT
290    FOR I = 1 TO N
300 READ M0(I) : PRINT TAB(4) M0(I) : NEXT  I : PRINT
320  PRINT TAB(6) " 3 - AVERAGE ADDITIONAL AREAS A(I) IN M0 DIAG." : PRINT
330  FOR I = 1 TO N / 2
340  READ A(I): PRINT A(I) :NEXT I : PRINT
360 PRINT TAB(6)" 4 - POSITION OF CENTROID  G FOR CG. OF A " : PRINT
370  FOR  I = 1 TO N / 2
380  READ G(I): PRINT  G(I) : NEXT  I : PRINT
385 PRINT TAB(6) " 5 - MAX. PARABOLIC ORDINATE W1(I,J) IN M1(I) - DIAGS. " : PRINT
390 FOR I = 1 TO  R
392    FOR J = 1 TO N/2
395 READ W1(I,J) : PRINT W1(I,J), :NEXT J : PRINT :NEXT I :   PRINT
400 PRINT  TAB(6)" 5 - MOMENTS M(I,J) DUE TO M(I) = 1 " :  PRINT
410   FOR I = 1 TO R
420    FOR J = 1 TO N
430  READ M(I,J) : PRINT M(I,J), : NEXT J
440   PRINT  : NEXT  I : PRINT
460 PRINT TAB(6) " 6 - ADDITIONAL DISP. COEF. T(I) DUE TO TIES":PRINT
470         FOR I = 1 TO R
480  READ T(I) : PRINT TAB(4) T(I) : NEXT I : PRINT
500 PRINT TAB(6) " 7 - ADDITIONAL DISP. COEF. U(I) DUE TO SETTLEMENT " : PRINT
510  FOR I = 1 TO R
520    READ U(I) : PRINT  U(I) : NEXT I : PRINT
550 PRINT TAB(6) " 8 - DISP. COIFS. D(I) DUE TO M0 & M1 DIAGS. " : PRINT
```

```
560  FOR I = 1 TO R
570   D(I) = 0
580    FOR K = 1 TO (N - 1)  STEP 2
590  D(I) = D(I) + L((K+1)/2) * ( M0(K)*M(I,K) /3  +  M0(K+1) * M(I,(K+1))/3 )
600 D(I)= D(I) +L((K+1)/2) *( M0(K)* M(I,(K+1))/6  + M0(K+1)* M(I,K )/6)
610 D(I)= D(I) +L((K+1)/2)* A((K+1)/2)*M(I,(K+1))* G((K+1)/2)
620 D(I) = D(I) + L((K+1)/2) * A((K+1)/2) * (1-G((K+1)/2))* M(I,K)
630 D(I) = D(I) +.8! *L((K+1)/2)* A((K+1)/2) * W1(I,(K +1)/2)
635 D(I) = D(I) +.3333! *L((K+1)/2)* (M0(K) + M0(K+1)) * W1(I,(K+1)/2)
640   NEXT K
650  D(I) = D(I) + U(I)
660    PRINT D(I) : NEXT I : PRINT
700 PRINT TAB(6)"9 - DISP. COEFS. S(I,J)  DUE TO M(I,J) " : PRINT
710  FOR I = 1 TO R
720   FOR J = 1  TO R
730     S(I,J) = 0
740   FOR K = 1 TO (N-1) STEP 2
750 S(I,J) = S(I,J) + L((K+1)/2)  *(M(I,K)*M(J,K)/3 + M(I,(K+1))*M(J,(K+1))/3)
760 S(I,J) = S(I,J) + L((K+1)/2)* (M(I,K)*M(J,(K+1))/6 + M(I,(K+1))* M(J,K)/6)
762 S(I,J) = S(I,J) +.3333! * L((K+1)/2) *(M(I,K) + M(I,(K+1))) * W1(J,((K+1) /2))
764 S(I,J) = S(I,J)+ .3333! *L((K+1)/2) *  (M(J,K)+M(J,(K+1)))* W1(I,((K+1)/2))
766 S(I,J) = S(I,J)+ .53333! * L((K+1)/2) * W1(I,((K +1)/2)) * W1(J , (K+1)/2)
770    NEXT K
780     IF J<>I THEN GOTO 800
790   S(I,J) =   S(I,J) + T(I)
800   NEXT J   :  NEXT I
820   FOR I = 1 TO R
830    FOR J = 1 TO R
840  PRINT S(I,J), : NEXT  J : PRINT
850   NEXT I : PRINT  : PRINT
880 PRINT TAB(6) " 10 - FLEXIBILITY MATRIX V = INVERS OF S " :  PRINT
910      FOR I = 1 TO R
920       FOR J = 1 TO R
930        Z(I,J) = S(I,J)
940                         NEXT  J  :          NEXT I: PRINT
950    IF R = 1 THEN  V(1,1) = 1 / S(1,1)  : PRINT V(1,1) :GOTO 1000
980  GOSUB 10000
1000  PRINT TAB(6)" 11 - REDUNDANT VALUES X (R) " : PRINT
1110   FOR I = 1 TO  R
1120      X(I) = 0
1140   FOR K = 1 TO   R
1150     X(I) = X(I) -V(I,K) * D(K)
1160       NEXT K : NEXT  I
1170  FOR I = 1 TO R
1180   PRINT TAB(4) X(I) :NEXT I : PRINT
1200   PRINT TAB(6)" 11 - FINAL B.MOMENTS BM    " : PRINT
1205       DIM  BM(N)
1210  FOR I = 1 TO N
1220      BM(I) = 0
```

```
1230  FOR J = 1 TO R
1240   BM(I) = BM(I)  + X(J) * M(J,I)
1250      NEXT  J
1260   BM(I) = BM(I) + M0(I)
1270    NEXT  I
1280 INPUT  " YOU WANT SCREEN (1) , PRINTER (2) , OR END (3) ? " , SP
1290  ON SP GOTO 1310 , 1360 , 1400
1310 FOR I = 1 TO (N-1) STEP 2
1320  PRINT TAB(4) BM(I) , BM(I+1)
1330    NEXT I : PRINT
1340  PRINT " THIS IS ALL NOW "
1350    GOTO 1280
1360 FOR I = 1 TO (N-1) STEP 2
1370 LPRINT TAB(4) BM(I) , BM(I+1)
1380  NEXT I : LPRINT
1390  LPRINT TAB(4)," THAT IS ALL NOW   -  GOOD BY "  :  LPRINT
1400  END
10000    FOR I = 1 TO R
10010      FOR J = (R+1) TO  2 * R
10020     Z(I,J) = 0
10030        NEXT J
10035     Z(I , R+I) = 1
10040        NEXT I
10050    P = 0
10060      P = P + 1
10070    C(P,P) = Z(P,P)
10080      FOR    J = 1 TO (2 * R)
10090    Z(P,J) = Z(P,J) / C(P,P)
10100       NEXT J
10110    FOR I = (P + 1)  TO  R
10120      C(I,P) = Z(I,P)
10130    FOR J = P  TO 2 * R
10140     Z(I,J) = Z(I,J) - C(I,P) * Z(P,J)
10150         NEXT J
10160           NEXT I
10170        IF P = 1  THEN 10060
10180     FOR I = 1 TO  (P-1)
10190      C(I,P) = Z(I,P)
10200    FOR J = P TO (2 * R)
10210     Z(I,J) = Z(I,J) - C(I,P) * Z(P,J)
10220       NEXT  J
10230         NEXT I
10240    IF P = R THEN GOTO 10260
10250     GOTO  10060
10260  FOR I = 1 TO R
10270    FOR J = 1 TO R
10280   V(I,J) = Z(I,R+J)
10290  PRINT V(I,J), : NEXT J
10300     PRINT   : NEXT  I : PRINT
10320        RETURN
```

```
10 PRINT TAB(4)," NAME OF PROGRAM  ----   VWPB ---- " : PRINT
20 PRINT TAB(4)," THIS PROGRAM GIVES THE SOLUTION BY  VW" : PRINT
25 PRINT TAB(6)" FOR FRAMES WITH SEVERAL CASES OF LOADING " : PRINT
30 PRINT "------------------------------------------------" : PRINT  : PRINT
40 REM  R = NOM. OF REDUNDANT VALUES
50 REM  N = NOM. OF MOMENTS AT THE ENDS OF MEMBERS
55 REM  B = NO. OF CASES OF LOADING
60 REM  M0(I,J) = MOMENTS IN THE M.S  DUE TO GIVEN CASES OF LOADING
70 REM  M1(I,J) = MOMENTS DUE TO X(I) = 1
90 REM  G(I,J) GIVES THE POSITION OF THE C.G. OF ADDITIONAL M0 DIAGS.
100 REM  A(I,J) = AVERAGE MOMENT IN THE ADDITIONAL M0 DIAG.
105 REM  W1(I,J) = MAX.MOMENT IN M(I,J) DIAGRAM "
110 REM  D0(I,J) = DISPLACEMENT COEFS. FROM M0 & M(I) DIAGS.
120 REM  S(I,J) = ................... FROM APPROPRIATE M(I,J) DIAGS.
140 REM  T(I,J) = ADDITIONAL VALUE TO S(I,J) DUE TO TIE IF ANY
150 REM  U(I,J) = .................. D(I)  DUE TO SETTLEMENT IF ANY
180  READ R , N , B
190  PRINT " R = " , R  ," N = ", N  , "B = " ,B
210        DIM L(N/2) , M0(B,N), W(B,N/2)   , G(B,N/2), Z(R,2*R)
220       DIM D(B,R)  , M(R,N) , S(R,R), C(R,R) , BM(N,B) , X(R,B)
225         DIM V(R,R) , D0(B,R) , W1(R,N/2)
240 PRINT  TAB(6)" 1 - ELASTIC LENGTHES L(I) "  :  PRINT
250       FOR I = 1 TO (N /2)
260 READ L(I) : PRINT TAB(4) L(I) : NEXT I : PRINT
280  PRINT TAB(6)" 2 - MOMENTS M0 IN  M.S. "  :  PRINT
290   FOR I = 1 TO B
295    FOR  J = 1 TO N
300 READ M0(I,J) :  PRINT  M0(I,J), : NEXT  J
310     PRINT : NEXT  I : PRINT
320  PRINT TAB(6) " 3 - AVERAGE ADDITIONAL AREAS A(I,J) IN M0 DIAGS.." : PRINT
330  FOR I = 1 TO B
335    FOR J = 1 TO N/2
340  READ A(I,J) : PRINT  A(I,J), :NEXT J : PRINT
350    NEXT I : PRINT
360 PRINT TAB(6)" 4 - POSITION OF CENTROID  G FOR CG. OF A " : PRINT
370  FOR  I = 1 TO B
375     FOR J = 1 TO N/2
380  READ G(I,J) : PRINT  G(I,J) ,: NEXT  J : PRINT
390      NEXT I : PRINT
392 PRINT TAB(6)" 5 - MAX. ORDINATE W1 IN M(I,J) DIAG.  "  : PRINT
394  FOR I = 1 TO  R
396    FOR J = 1 TO N/2
398  READ W1(I,J) : PRINT  W1(I,J), : NEXT J : PRINT
400    NEXT I :  PRINT
405 PRINT TAB(6)" 6 _ ENTER ORDINATES OF M(I,J) DIAGRAMS  : : PRINT
410  FOR I = 1 TO R
420    FOR J = 1 TO N
430  READ M(I,J) : PRINT M(I,J), : NEXT J
```

```
440   PRINT  : NEXT  I : PRINT
460 PRINT TAB(6) " 7 - ADDITIONAL DISP. COEF. U(I) DUE TO SETLEMENTT" :  PRINT
470           FOR I = 1 TO R
480 READ U(I)   : PRINT   U(I)   :  NEXT I : PRINT
500 PRINT TAB(6) " 8 - ADDITIONAL DISP. COEF. T(I)  DUE TO TIES " : PRINT
510  FOR I = 1 TO R
520   READ T(I)     : PRINT  T(I)    : NEXT I : PRINT
550 PRINT TAB(6) " 9 - DISP. COIFS. D0(I,J) DUE TO M0 & M1 DIAGS. " : PRINT
560  FOR I = 1 TO B
565    FOR J = 1 TO R
570   D(I,J) = 0
580    FOR K = 1 TO (N - 1)  STEP 2
590  D(I,J) = D(I,J) + L((K+1)/2) * ( M0(I,K) * M(J,K) /3  +  M0(I,(K+1)) * M(J,(K+1))/3 )
600 D(I,J)= D(I,J) +L((K+1)/2) *(M0(I,K)* M(J,(K+1))/6  + M0(I,(K+1))* M(J,K )/6)
610 D(I,J)= D(I,J) +L((K+1)/2)* A(I,(K+1)/2)*M(J,(K+1))* G(I,(K+1)/2)
620 D(I,J) =D(I,J) + L((K+1)/2) * A(I,(K+1)/2) * (1-G(I,(K+1)/2))* M(J,K)
630 D(I,J) = D(I,J)+.8! *L((K+1)/2) * A(I,(K+1)/2)  * W1( J,(K+1)/2)
635 D(I,J) = D(I,J) + .3333! * L((K+1)/2) *  (M0(I,K) + M0(I,(K+1)))* W1(J,(K+1)/2)
640   NEXT K
650  D(I,J) = D(I,J) + U(I)
654    NEXT J : NEXT I
656 FOR I = 1 TO B
658   FOR J = 1 TO R
660    D0(I,J) = D(I,J) :NEXT J: NEXT I
662  FOR I = 1 TO B
664  FOR J = 1 TO R
666  PRINT D0(I,J),: NEXT J  : PRINT
668   NEXT I  :  PRINT
700 PRINT TAB(6) " 10 - DISPLACEMENT COEFS. S(I,J) DUE TO M(I,J) " : PRINT
710  FOR I = 1 TO R
720   FOR J = 1  TO R
730     S(I,J) = 0
740  FOR K = 1 TO (N-1) STEP 2
750 S(I,J) = S(I,J) + L((K+1)/2)  * (M(I,K)* M(J,K)/3 + M(I,(K+1))*M(J,(K+1))/3)
760 S(I,J) = S(I,J) + L((K+1)/2)* (M(I,K)*M(J,(K+1))/6  +M(I,(K+1)) * M(J,K)/6)
762 S(I,J) = S(I,J) + .3333! * L((K+1)/2) * ( M(I,K) + M(I,(K+1))) * W1(J,(K+1)/2)
764 S(I,J) = S(I,J) + .3333! * L((K+1)/2) * (M(J,K) + M(J,(K+1))) * W1(I,(K+1)/2)
766 S(I,J) = S(I,J) + .5333! * L((K+1)/2) * W1(I,(K+1)/2) * W1(J,(K+1)/2)
770   NEXT K
780    IF J<>I THEN GOTO 800
790  S(I,J) =  S(I,J) + T(I)
800  NEXT J  :  NEXT I
810  FOR I = 1 TO R
820     FOR J = 1 TO R
830 PRINT S(I,J),: NEXT J : PRINT
840 NEXT I : PRINT
880 PRINT TAB(6) " 10 - FLEXIBILITY MATRIX V = INVERS OF S " :  PRINT
910     FOR I = 1 TO R
```

```
920       FOR J = 1 TO R
930         Z(I,J) = S(I,J)
940             NEXT  J  :     NEXT I: PRINT
950    IF R = 1 THEN  V(1,1) = 1 / S(1,1)  : PRINT V(1,1) :GOTO 1000
980 GOSUB 10000
1000 PRINT TAB(6)" 11 - REDUNDANT VALUES X (I,J)  " : PRINT
1110    FOR I = 1 TO  R
1115      FOR J = 1 TO B
1120        X(I,J) = 0
1140    FOR K = 1 TO   R
1150      X(I,J) = X(I,J) -V(I,K) * D0(J,K)
1160        NEXT K : NEXT J : NEXT I
1170   FOR I = 1 TO R
1175          FOR J = 1 TO B
1180       PRINT X(I,J),: NEXT J : PRINT
1190         NEXT I : PRINT
1200     PRINT TAB(6)" 11 - FINAL B.MOMENTS BM    " : PRINT
1210 FOR I = 1 TO N
1215      FOR J = 1 TO B
1220        BM(I,J) = 0
1230    FOR K = 1 TO R
1240     BM(I,J)  = BM(I,J)   + M(K,I) * X(K,J)
1250        NEXT  K
1260     BM(I,J)  = BM(I,J) +  M0(J,I)
1270      NEXT  J : NEXT  I
1272 INPUT "YOU WANT SCREEN (1) , PRINTER (2) , OR END (3) " , SPE
1274 ON SPE  GOTO 1280 , 1340 , 1400
1280 FOR I = 1 TO N
1285     FOR J = 1 TO B
1290   PRINT  BM(I,J), : NEXT J  :    PRINT
1310     NEXT I : PRINT
1320   PRINT " THIS IS ALL NOW " : GOTO 1272
1340   FOR I = 1 TO N
1350    FOR J = 1 TO B
1360   LPRINT  BM(I,J),:  NEXT J :  LPRINT
1370     NEXT I :  LPRINT
1380 LPRINT  " THAT IS ALL NOW  _  GOOD BY  "
1400     END
10000     FOR I = 1 TO R
10010       FOR J = (R+1) TO  2 * R
10020         Z(I,J) = 0
10030           NEXT J
10035      Z(I , R+I) = 1
10040           NEXT I
10050     P = 0
10060         P = P + 1
10070     C(P,P) = Z(P,P)
10080       FOR    J = 1 TO (2 * R)
```

```
10090    Z(P,J) = Z(P,J) / C(P,P)
10100       NEXT J
10110    FOR I = (P + 1)  TO  R
10120      C(I,P) = Z(I,P)
10130    FOR J = P  TO 2 * R
10140     Z(I,J) = Z(I,J) - C(I,P) * Z(P,J)
10150          NEXT J
10160             NEXT I
10170        IF P = 1  THEN 10060
10180     FOR I = 1 TO  (P-1)
10190      C(I,P) = Z(I,P)
10200    FOR J = P TO (2 * R)
10210     Z(I,J) = Z(I,J) - C(I,P) * Z(P,J)
10220       NEXT  J
10230          NEXT I
10240    IF P = R THEN GOTO 10260
10250     GOTO  10060
10260  FOR I = 1 TO R
10270    FOR J = 1 TO R
10280    V(I,J) = Z(I,R+J)
10290 PRINT V(I,J), : NEXT J
10300     PRINT   : NEXT  I : PRINT
10320       RETURN
```

```
2990 PRINT " NAME        DATAVW  "
3000 PRINT " ENTER DATA FOR  R , N "
3020 PRINT " ENTER DATA FOR ELASTIC LENGTHES L(I) "
3040 PRINT " ENTER DATA B.MOMENTS M0(I) IN M.S. "
3080 PRINT " ENTER DATA FOR ADDITIONAL AVERAGE ORDINATE IN A(I) "
3120 PRINT " ENTER DATA FOR G: CENTROID OF ADDIT. AREA A(I) "
3160 PRINT " ENTER DATA FOR M(I,J) DUE TO M(I) = 1 "
3240 PRINT " ENTER DATA FOR ADDITIONAL DISP. COEFS. U(I) "
3260 PRINT " ENTER DATA FOR ADDITIONAL DISP. COEFS. T(I) "
```

```
2990 PRINT "  NAME  -----------------  DATAVWP     "
3000 PRINT " ENTER DATA FOR  R , N "
3020 PRINT " ENTER DATA FOR ELASTIC LENGTHES L(I) "
3040 PRINT " ENTER DATA B.MOMENTS M0(I) IN M.S. "
3080 PRINT " ENTER DATA FOR ADDITIONAL AVERAGE ORDINATE IN A(I) "
3120 PRINT " ENTER DATA FOR G: CENTROID OF ADDIT. AREA A(I) "
3140 PRINT " ENTER DATA FOR MAX. ORDINATE  W1(I,J), IN M1(I) - DIAGS> "
3260 PRINT " ENTER DATA FOR M1(I,J) "
3360 PRINT " ENTER DATA FOR ADDITIONAL DISP. COEFS. T(I) "
3460 PRINT " ENTER DATA FOR ADDITIONAL DISP. COEFS. U(I) "
```

```
2990 PRINT  " NAME ------------   DATVWPB      "
3000 PRINT " ENTER DATA FOR  R , N , B "
3020 PRINT " ENTER DATA FOR ELASTIC LENGTHES L(I) "
3040 PRINT " ENTER DATA B.MOMENTS M0(I,J) IN M.S. "
3080 PRINT " ENTER DATA FOR ADDITIONAL AVERAGE ORDINATE IN A(I) "
3120 PRINT " ENTER DATA FOR G: CENTROID OF ADDIT. AREA A(I,J) "
3160 PRINT "ENTER DATA FOR W1(I,J)
3250 PRINT " ENTER DATA FOR MOMENTS M(I,J) FOR X(I) = 1  "
3360 PRINT " ENTER DATA FOR ADDITIONAL DISP. T(I,J) DUE TO TIES "
3440 PRINT "ENTER DATA F0R ADDITIONAL DISP.U(I,J)DUE TO SETTLEMENTS "
```

```
2990 PRINT  " NAME ------------ DATAVW02 "
3000 PRINT " ENTER DATA FOR  R , N "
3010 DATA  7 , 20
3020 PRINT " ENTER DATA FOR ELASTIC LENGTHES L(I) "
3030 DATA 4 , 5.4975 , 2 , 1.5 , 3.354
3035 DATA 3.354 , 1.5 , 2 , 5.4975 , 4
3040 PRINT " ENTER DATA B.MOMENTS M0(I) IN M.S. "
3050 DATA 0 , 0 , 0 , -72 , -36 , 0 , -36 , -36 , -36 , 0
3060 DATA 0 , -36 , -36 , -36 , 0 , 0 , -36 , 0 , 0 , 0
3080 PRINT " ENTER DATA FOR ADDITIONAL AVERAGE ORDINATE IN A(I) "
3090 DATA 0 , 10.6667 , 0 , 0 , 6
3100 DATA 6 , 0 , 0 , 10.6667 , 0
3120 PRINT " ENTER DATA FOR G: CENTROID OF ADDIT. AREA A(I) "
3130 DATA 0 , .5 , 0 , 0 , .5
3135 DATA .5 , 0 , 0 , .5 , 0
3260 PRINT " ENTER DATA FOR M1(I,J) "
3270 DATA 0 , -4 , -4 , -6 , -6 , 0 , 0 , 0 , 0 , 0
3275 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0
3280 DATA   0 , 0 , 0 , -1 , -1 , -1 , 0 , 0 , 0 , 0
3285 DATA   0 , 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0
3290 DATA 0 , 0 , 0 , 12 , 6 , 0 , 6 , 3 , 3 , 0
3295 DATA 0 , 3 , 3 , 6 , -6 , 0 , 12 , 0 , 0 , 0
3300 DATA     0 , 0 , 0 , 6 , 0 , 0 , 6 , 6 , 6 , 0
3305 DATA     0 , -6 , -6 , -6 , 0 , 0 , -6 , 0 , 0 , 0
3310 DATA 0 , 0 , 0 , -1 , 0 , 0 , -1 , -1 , -1 , -1
3315 DATA -1 , -1 , -1 , -1 , 0 , 0 , -1 , 0 , 0 , 0
3320 DATA    0 , 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0
3325 DATA    0 , 0 , 0 , 0 , 1 , 1 , -1 , 0 , 0 , 0
3330 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0
3335 DATA 0 , 0 , 0 , 0 , 6 , 0 , -6 , -4 , -4 , 0
3360 PRINT " ENTER DATA FOR ADDITIONAL DISP. COEFS. T(I) "
3370 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0
3460 PRINT " ENTER DATA FOR ADDITIONAL DISP. COEFS. U(I) "
3470 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0 , 0
```

```
2990 PRINT " NAME  ------------ DATAVWP1
3000 PRINT " ENTER DATA FOR  R , N "
3010 DATA  7 , 14
3020 PRINT " ENTER DATA FOR ELASTIC LENGTHES L(I) "
3030 DATA 4 , 12 , 2 , 6 , 2 , 12 , 4
3040 PRINT " ENTER DATA B.MOMENTS M0(I) IN M.S. "
3050 DATA 0 , 18 , 18 , 0 , 0 , 0 , 0
3055 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3080 PRINT " ENTER DATA FOR ADDITIONAL AVERAGE ORDINATE IN A(I) "
3090 DATA 3 , 24 , 0 , 81 , 0 , 0 , 0
3120 PRINT " ENTER DATA FOR G: CENTROID OF ADDIT. AREA A(I) "
3125 DATA .5 , .5 , 0 , .5 , 0 , 0 , 0
3140 PRINT " ENTER DATA FOR MAX. ORDINATE  W1(I,J), IN M1(I) - DIAGS> "
3145 DATA 0 , 3 , 0 , 0 , 0 , 0 , 0
3150 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3155 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3160 DATA 0 , 0 , 0 , 4.5 , 0 , 0 , 0
3165 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3170 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3175 DATA 0 , 0 , 0 , 0 , 0 , .5 , 0
3260 PRINT " ENTER DATA FOR M1(I,J) "
3275 DATA 0 , 6 , 6 , 0 , 6 , 0 , -6
3280 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3285 DATA     0 , 0 , 0 , 1 , 0 , 0 , 1
3290 DATA     0 , 0 , 0 , 0 , 0 , 0 , 0
3295 DATA 0 , 0 , 0 , 0 , -1 , -1 , 1
3300 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3305 DATA     0 , 0 , 0 , 0 , -6 , 0 , 6
3310 DATA     0 , 6 , 0 , -6 , 0 , 0 , 0
3315 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3320 DATA 1 , 0 , 0 , 1 , 0 , 0 , 0
3330 DATA     0 , 0 , 0 , 0 , 0 , 0 , 0
3335 DATA     0 , -1 , -1 , 1 , 0 , 0 , 0
3340 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3345 DATA 0 , -1 , 0 , 1 , 1 , 1 , 0
3360 PRINT " ENTER DATA FOR ADDITIONAL DISP. COEFS. T(I) "
3370 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3460 PRINT " ENTER DATA FOR ADDITIONAL DISP. COEFS. U(I) "
3470 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
```

```
2990 PRINT " NAME ----------------- DATVWPB1 "
3000 PRINT " ENTER DATA FOR R , N , B "
3010 DATA 7 , 14 , 3
3020 PRINT " ENTER DATA FOR ELASTIC LENGTHES L(I) "
3030 DATA 4 , 12 , 2 , 6 , 2 , 12 , 4
3040 PRINT " ENTER DATA B.MOMENTS M0(I,J) IN M.S. "
3045 DATA 0 , 18 , 18 , 0 , 0 , 0 , 0
3048 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3055 DATA    0 , 0 , 0 , 0 , 0 , 0 , 0
3058 DATA    0 , 0 , 0 , 0 , 0 , 0 , 0
3065 DATA 0 , 18 , 18 , 0 , 0 , 0 , 0
3068 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3080 PRINT " ENTER DATA FOR ADDITIONAL AVERAGE ORDINATE IN A(I) "
3085 DATA 3 , 0 , 0 , 0 , 0 , 0 , 0
3088 DATA 0 , 24 , 0 , 81 , 0 , 0 , 0
3095 DATA 3 , 24 , 0 , 81 , 0 , 0 , 0
3120 PRINT " ENTER DATA FOR G: CENTROID OF ADDIT. AREA A(I,J) "
3125 DATA .5 , 0 , 0 , 0 , 0 , 0 , 0
3130 DATA 0 , .5 , 0 , .5 , 0 , 0 , 0
3135 DATA .5 , .5 ,0 , .5 , 0 , 0 , 0
3160 PRINT "ENTER DATA FOR W1(I,J)
3165 DATA 0 , 3 , 0 , 0 , 0 , 0 , 0
3170 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3175 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3185 DATA 0 , 0 , 0 , 4.5 , 0 , 0 , 0
3190 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3195 DATA  0 , 0 , 0 , 0 , 0 , 0 , 0
3200 DATA 0 , 0 , 0 , 0 , 0 , .5 , 0
3250 PRINT " ENTER DATA FOR MOMENTS M(I,J) FOR X(I) = 1 "
3260 DATA 0 , 6 , 6 , 0 , 6 , 0 , -6
3265 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3270 DATA    0 , 0 , 0 , 1 , 0 , 0 , 1
3275 DATA    0 , 0 , 0 , 0 , 0 , 0 , 0
3280 DATA 0 , 0 , 0 , 0 , -1 , -1 , 1
3285 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3290 DATA    0 , 0 , 0 , 0 , -6 , 0 , 6
3295 DATA    0 , 6 , 0 , -6 , 0 , 0 , 0
3300 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3305 DATA 1 , 0 , 0 , 1 , 0 , 0 , 0
3310 DATA    0 , 0 , 0 , 0 , 0 , 0 , 0
3315 DATA    0 , -1 , -1 , 1 , 0 , 0 , 0
3320 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3325 DATA 0 , -1 , 0 , 1 , 1 , 1 , 0
3360 PRINT " ENTER DATA FOR ADDITIONAL DISP. T(I,J) DUE TO TIES "
3400 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3410 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3420 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3440 PRINT "ENTER DATA F0R ADDITIONAL DISP.U(I,J)DUE TO SETTLEMENTS "
3450 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3460 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
3470 DATA 0 , 0 , 0 , 0 , 0 , 0 , 0
```